

Battling Tanks

Copyright © Sebree's Computing 1980

INSTRUCTIONS TO BATTLING TANKS

MEMORY REQUIRED- 8K

Battling Tanks is a game that requires 2 joysticks and two players. The game is based on the idea that the 2 players are looking down on the battle-grounds, seeing their tanks from above and maneuvering them from there. The general rules of the game are that the two players are trying to blow each others tank apart with the useful missile turret they are armed with. The tanks rotate about their axis in 45 degree increments. You may rotate your tank around clockwise by pushing your joystick to the right. To move your tank the opposite direction, just push to the left. To move forward, just push the joystick forward. To back-up in ANY position, just pull back on the joystick.

In order to gain the most points, you must maneuver your tank to a position which is aimed at the opposing player, and fire your cannon by pressing your trigger. You must NOT be TOO CLOSE to the other tank when you shoot or your missile will go directly over the other player. If you hit the other tank, you will gain points. The screen will now show the score at the bottom. If you hit any key on the computer, the game will continue. There are only a few bytes of memory left which are used up within loops, so that the program could not be expanded beyond the point that it is now. Thus, at the beginning of the game, both players should agree upon a total score that one player must reach in order to win. The score will be displayed at the end of each round. The program was done entirely in BASIC but still executes quite fast. The tanks rotate around at a fairly fast speed, so if the opposing player shoots at you, you still have a chance to get away from his advancing shot.

At the start of the game, the playfield (battlegrounds) will be set-up. The brown lines will be considered fences, and the green triangular figures will be considered to be barricades. You are capable of running over any one (or groups) of these objects, but your advancing speed will suffer. Whenever you run over something, OR, you are surrounded by foreign objects, your tank will slow down. Once you are in the clear, you will notice that your tank will run faster. A good strategy is to get in a cleared area to move away from the other player's shots while trying to fire at him at the same time. You may shoot as much as you wish. However, if you fire one shot and immediately fire a second shot, your first shot will disappear, being erased by the second shot. If one of your shots hits a fence or barricade, it will disappear. Your tanks may remain only within the screen area. If you try to run off the screen, you will be bumped 2 spaces backwards. Try not to get into a corner and be an easy target. Remember, the motto is to HAVE FUN!!!

```

1 REM BAT TNK:(C)SEBREE'S
10 B1=0:B2=0
20 X=0:Y=0:F=3.5:GRAPHICS 21:POKE 712,16:POKE 710,88
150 COLOR 1:FOR A=1 TO 14:AA=RND(0)*64:L=RND(0)*24:LL=RND(0)*64:PLOT AA,L:DRAWTO
AA+AA,L:PLOT LL,L:DRAWTO LL,L+AA:NEXT A
170 COLOR 2:FOR A=1 TO 9:X=RND(0)*69:Y=RND(0)*36:PLOT X,Y:DRAWTO X+9,Y:DRAWTO X,
Y+9:DRAWTO X+4,Y:DRAWTO X,Y+4:NEXT A
200 A=0:X=10:Y=10:JX=70:JY=30:R=0:S=0:AY=4
240 COLOR 4:GOSUB 800:TRAP 250
245 D=0:LOCATE X+7,Y,L:D=L:LOCATE X-8,Y,L:D=D+L:LOCATE X,Y+8,L:D=D+L:LOCATE X,Y-
7,L:D=D+L:C=1:IF D>0 THEN C=0.5
250 JS=STICK(0):IF JS=7 OR JS=5 OR JS=6 THEN A=A+1
260 POKE 77,0:IF JS=11 OR JS=10 OR JS=9 THEN A=A-1
270 IF JS=13 OR JS=9 OR JS=5 THEN F=-C
275 IF STRIG(0)=0 THEN GOSUB 410
277 IF R=1 THEN GOSUB 450
280 IF JS=14 OR JS=6 OR JS=10 THEN F=C
285 SOUND 0,65,4,4:IF JS=15 THEN F=0:SOUND 0,75,4,4
340 COLOR 3:GOSUB 780:X1=X:Y1=Y:AX=A:AY=A:JX=X:JY=Y:COLOR 4:GOSUB 800:TRAP 350
345 D=0:LOCATE X+7,Y,L:D=L:LOCATE X-8,Y,L:D=D+L:LOCATE X,Y+8,L:D=D+L:LOCATE X,Y-
7,L:D=D+L:C=1:IF D>0 THEN C=0.5
350 JS=STICK(1):IF JS=7 OR JS=5 OR JS=6 THEN A=A+1
360 IF JS=11 OR JS=10 OR JS=9 THEN A=A-1
370 IF JS=13 OR JS=9 OR JS=5 THEN F=-C
375 IF STRIG(1)=0 THEN GOSUB 510
377 IF S=1 THEN GOSUB 550
380 IF JS=14 OR JS=6 OR JS=10 THEN F=C
385 SOUND 1,124,2,4:IF JS=15 THEN F=0:SOUND 1,183,2,4
400 COLOR 3:GOSUB 780:AY=A:JX=X:JY=Y:X=X1:Y=Y1:A=AX:GOTO 240
410 SOUND 3,129,6,12:R=1:JX1=X:JY1=Y:G=0:H=0:IF A=0 THEN G=1
415 IF A=0 OR A=1 OR A=7 THEN G=3:JX1=JX1+8
420 IF A=3 OR A=4 OR A=5 THEN G=-3:JX1=JX1-8
425 IF A=1 OR A=2 OR A=3 THEN H=3:JY1=JY1+8
430 SOUND 3,0,0,0:IF A=5 OR A=6 OR A=7 THEN H=-3:JY1=JY1-8
450 TRAP 900:COLOR 4:PLOT JX1,JY1:COLOR 2:LOCATE JX1+6,JY1+H,L:IF L>0 THEN 900
460 JX1=JX1+G:JY1=JY1+H:PLOT JX1,JY1
470 RETURN
510 SOUND 3,129,6,12:S=1:JX2=X:JY2=Y:I=0:K=0:IF A=0 THEN G=1
515 IF A=0 OR A=1 OR A=7 THEN I=3:JX2=JX2+8
520 IF A=3 OR A=4 OR A=5 THEN I=-3:JX2=JX2-8
525 IF A=1 OR A=2 OR A=3 THEN K=3:JY2=JY2+8
530 SOUND 3,0,0,0:IF A=5 OR A=6 OR A=7 THEN K=-3:JY2=JY2-8
550 TRAP 900:COLOR 4:PLOT JX2,JY2:COLOR 2:LOCATE JX2+I,JY2+K,L:IF L>0 THEN 950
560 JX2=JX2+I:JY2=JY2+K:PLOT JX2,JY2:RETURN
780 IF A=0 OR A=1 OR A=7 THEN X=X+F
782 IF A=3 OR A=4 OR A=5 THEN X=X-F
783 IF A=1 OR A=2 OR A=3 THEN Y=Y+F
785 IF A=5 OR A=6 OR A=7 THEN Y=Y-F
788 IF X>71 THEN X=X-2
790 IF X<8 THEN X=X+2
793 IF Y>39 THEN Y=Y-2
796 IF Y<8 THEN Y=Y+2
800 IF A=0 THEN PLOT X,Y:DRAWTO X+7,Y:PLOT X-1,Y+1:DRAWTO X+4,Y+1:PLOT X-1,Y-1:D
RAWTO X+4,Y-1:RETURN
805 IF A=1 THEN PLOT X,Y:DRAWTO X+5,Y+5:PLOT X+1,Y-1:DRAWTO X+4,Y+2:PLOT X-1,Y+1
:DRAWTO X+2,Y+4:RETURN
810 IF A=2 THEN PLOT X,Y:DRAWTO X,Y+7:PLOT X+1,Y-1:DRAWTO X+1,Y+4:PLOT X-1,Y-1:D
RAWTO X-1,Y+4:RETURN
815 IF A=3 THEN PLOT X,Y:DRAWTO X-5,Y+5:PLOT X-1,Y-1:DRAWTO X-4,Y+2:PLOT X+1,Y+1
:DRAWTO X-2,Y+4:RETURN
820 IF A=4 THEN PLOT X,Y:DRAWTO X-7,Y:PLOT X+1,Y-1:DRAWTO X-4,Y-1:PLOT X+1,Y+1:D
RAWTO X-4,Y+1:RETURN
825 IF A=5 THEN PLOT X,Y:DRAWTO X-5,Y-5:PLOT X+1,Y-1:DRAWTO X-2,Y-4:PLOT X-1,Y+1
:DRAWTO X-4,Y-2:RETURN
830 IF A=6 THEN PLOT X,Y:DRAWTO X,Y-7:PLOT X+1,Y+1:DRAWTO X+1,Y-4:PLOT X-1,Y+1:D
RAWTO X-1,Y-4:RETURN
835 IF A=7 THEN PLOT X,Y:DRAWTO X+5,Y-5:PLOT X-1,Y-1:DRAWTO X+2,Y-4:PLOT X+1,Y+1
:DRAWTO X+4,Y-2:RETURN
840 IF A<0 THEN A=A+8
845 IF A>7 THEN A=0
850 GOTO 800
900 IF L=3 THEN POP :B1=B1+1:X=JX:Y=JY:GOTO 970
910 R=0:RETURN
950 IF L=3 THEN POP :B2=B2+1:X=X1:Y=Y1:GOTO 970
960 S=0:RETURN
970 COLOR 3:FOR L=143 TO 96 STEP -1:SOUND 0,RND(0)*L,8,8:PLOT X,Y:POKE 710,L:DR
AWTO RND(0)*79,RND(0)*47:NEXT L:POKE 710,13
980 SOUND 0,0,0,0:GRAPHICS 37:? "SCORE- PLR 1= ";B1;" : PLR 2= ";B2;" :POKE 764
,255:POP
990 IF PEEK(764)=255 THEN 990
995 GOTO 20
    
```